NASA Technical Memorandum 86324

Computer Subroutines for Estimation of Human Exposure to Radiation in Low Earth Orbit

Francis A. Cucinotta and John W. Wilson

FEBRUARY 1985

NASA Technical Memorandum 86324

Computer Subroutines for Estimation of Human Exposure to Radiation in Low Earth Orbit

Francis A. Cucinotta
Old Dominion University
Norfolk, Virginia

John W. Wilson
Langley Research Center
Hampton, Virginia

National Aeronautics
and Space Administration

Scientific and Technical
Information Branch

1985

Summary

Computer subroutines to calculate human exposure to trapped radiations in low Earth orbit (LEO) on the basis of a simple approximation of the human geometry by spherical shell shields of varying thickness are presented and detailed. The subroutines calculate the dose to critical body organs and the fraction of exposure limit reached as a function of altitude of orbit, degree of inclination, shield thickness, and days in mission. Exposure rates are compared with current exposure limits.

Introduction

With the advent of the Space Transportation System, there is rapid advancement in utilization of space in low Earth orbit (LEO). Principal interests in LEO are observation satellites, large space antennas, and a permanently manned space station. Increasing power requirements to promote manned capability and space industrialization are demanding large area solar arrays in addition to large components of living and working quarters. The net effect is increased atmospheric drag requiring higher orbital altitudes and greater radiation exposure. Furthermore, greater demands are being placed on human performance as a result of the high levels of extravehicular activity (EVA) associated with erectable structures and a manned space station.

In planning such missions, it is necessary to consider the impact of radiation exposure on mission activity. This report describes computer subroutines to calculate radiation exposure rates to various organs of the human body and to compare these rates with exposure limits. (See appendix A.) The computer code uses simple geometrical models of the human body and of the spacecraft to provide first-order estimates of limits for planning purposes. The models are based on time-averaged exposure rates without regard to important time variations in exposure.

Symbols

a, b, d	coefficients used in calculations
B.F.O.	blood forming organ
EVA	extravehicular activity
h	altitude, km
r	dosimeter radius, g/cm^2
z	thickness, g/cm^2

Spacecraft Shielding and Environmental Data

The complex geometric structure of all spacecraft makes specific exposure relations within their interior difficult to define. Various approximate methods have been developed over the years, which have resulted in great simplification (refs. 1 to 4). It is assumed that a large habitat can be approximated by a spherical shell with the astronaut at the center. This is a maximum exposure for such a spherical configuration.

In the present calculations, only radiations trapped in the Earth's magnetic field are considered. The effect of ignoring other sources of radiation is discussed in reference 5. The trapped particle fluence is taken from a compilation of data (ref. 6) derived from the AE4 electron model for inner-zone electrons, AE5 outer-zone electron models which intersect low altitudes at high latitudes, and AP5, AP6, and AP7 proton models which are combined as low-, medium-, and high-energy protons for solar maximum. More recent data are obtainable from the National Space Science Data Center (NSSDC) but were not available for this study. The new data set is within the factor 2 of uncertainty of the inner-zone model.

The computer code SHIELDOSE of Seltzer from the National Bureau of Standards (ref. 7) is used to convert the trapped radiation fluence data to one-half dose at the center of a solid aluminum sphere. The data are for 42 shield thicknesses ranging from 0.03 g/cm^2 to 30.0 g/cm^2 for altitudes of 200 km at 30°, 60°, and 90° inclination and 400, 600, 800, and 1000 km at 0°, 30°, 60°, and 90° inclination. The data calculated by SHIELDOSE are converted to full dose in rads per day by the computer subroutine READTPE.

An interpolation procedure is performed by the computer subroutine DOSECLC to calculate doses as a function of altitude. To interpolate in altitude, it is assumed that the dose is proportional to a power of the altitude

$$D = bh^a \tag{1}$$

where D is the dose to be calculated at the altitude h, and a and b are defined as follows:

$$a = \frac{\log(D_1/D_2)}{\log(h_1/h_2)} \tag{2}$$

$$b = \frac{D_1}{h_1^a} \tag{3}$$

In equations (2) and (3), h_1 and h_2 are the first altitudes existing in the data base below and above the point of interpolation, respectively, and D_1 and D_2 are the corresponding doses.

No data existed in the data base for an altitude of 200 km at 0° inclination. To extrapolate into the region between 200 km and 400 km at this inclination, the following approximation is used:

$$D = 100^P \; D(h = 400) \tag{4}$$

where

$$P = \frac{h - 400}{100} \tag{5}$$

In equation (4), $D(h = 400)$ is the dose at 400 km at 0° inclination, and D is the dose to be calculated at the altitude h.

To interpolate as a function of shield thickness, the subroutine IUNI of the mathematical subroutine library of the Langley Central Scientific Computer Complex is used. Subroutine IUNI uses a first-order Lagrangian interpolation. Also, because the data base contained only four values of degree of inclination, no interpolation was attempted over this variable.

Astronaut Self-Shielding

The human body is a complicated geometric arrangement, and the specific organs of interest are likewise distributed in complex geometric patterns. Detailed man models have been derived (ref. 4) and substantially improved (ref. 8). To approximate the dose to various body organs, the work of Billings and Langley (ref. 9), which uses a simple spherical shell model of critical body organs, is utilized. This model is represented by spherical shell thickness equivalent to the depth of the organ and a coefficient representing the amount of radiation incident on the organ in question. The present model used the minimum-number proton dosimeters parameters (table 3 of ref. 9) except for the skin dose, which used the minimum-error parameters (table 2 of ref. 9). The skin dose is approximated by a dosimeter radius

$$r = \begin{cases} z/4 & (z \leq 8 \text{ g/cm}^2) \\ 2 & (z > 8 \text{ g/cm}^2) \end{cases} \tag{6}$$

with coefficient

$$C(z) = a + be^{-\alpha z} \tag{7}$$

where z is the vehicle shield thickness. The remaining organs are correspondingly approximated for a constant r shown in table 1 along with the coefficients a, b, and α, used in the present calculations.

Method of Calculation

In calculating the dose to specific body organs, the human body geometry and spacecraft geometry are combined according to the joint probability distribution (ref. 9), which for our simplified geometry becomes

$$D_{\text{organ}} = C_{\text{organ}}(z)\ D_{\text{sphere}}(r_{\text{organ}} + Z) \tag{8}$$

where $C_{\text{organ}}(z)$ is the coefficient calculated by equation (7) for the specific body organ, $D_{\text{sphere}}(x)$ is the dose in the center of an aluminum sphere of radius x; r_{organ} is the corresponding organ radius (table 1), and z is the spacecraft shield thickness assumed to be a spherical shell with the dose point at the center. An example of a specific shield is associated with each thickness shown in table 2 as noted.

The calculations are made as a function of altitude of orbit, degree of inclination, thickness of spacecraft, and days in mission. These variables are specified by the user and must be passed to the subroutine DOSECLC from the user's main program. Limitations on the values of these user-specified variables are a minimum of 200 km and a maximum of 1000 km for the altitude of orbit, values of 0°, 30°, 60°, and 90° for degree of inclination, and a minimum of 0.03 g/cm^2 and a maximum of 24.5 g/cm^2 for thickness of spacecraft.

Radiation exposure constraints are discussed in reference 5. Exposure limits were available for 30-, 60-, 365-, and 3650-day periods (table 3). Exposure limits have been revised for the blood forming organs (B.F.O.), skin, and lens, and these values are used in the present calculations. For the testes, we use the values of reference 10. The fraction of exposure limit reached for a mission is calculated by the subroutines for the first exposure period above the number of days specified by the user. An exception is for missions longer than 365 days. In these cases, multiples of the 365-day limit are used.

Sample Calculations

Sample calculations using the computer subroutines for a 90-day mission are shown in tables 4 and 5. For a given shield thickness, exposure limits are approached and in some cases exceeded for the higher altitudes, depending on the body organ and degree of inclination.

A main program utilizing the computer subroutines for a 90-day mission with 8 hours of EVA every 10 days is shown in appendix B. The results generated are shown in table 6. The shielding values for time in EVA are given in table 2. The shielding for the lens is taken as a space helmet, and shielding for the B.F.O., skin, and testes is taken as a space suit. The increase in the fraction of exposure limit caused by the time in EVA can be seen by comparing these results with the corresponding values for a mission without EVA in table 4(a). The EVA causes less than a 3-percent increase in the exposure limit for the B.F.O., lens, and testes, and an increase of approximately 20 percent for the skin.

Concluding Remarks

A set of computer subroutines have been developed to estimate long-term time-averaged human exposure in low Earth orbit (LEO), and the use of these subroutines has been explained. Users of the subroutines should be

mindful of the limitations of the simple geometric models of the human body and of the spacecraft, as well as the inherent uncertainties of the environmental models (approximately a factor of 2). Results of these subroutines should be interpreted in the context of current radiation constraints. Time variations in exposure rates have not been taken into account and are expected to be of vital importance during extravehicular activity operations as a means of reducing exposure.

Langley Research Center
National Aeronautics and Space Administration
Hampton, VA 23665
November 14, 1984

Appendix A

Program Listing

The computer subroutines given in this appendix were developed for the present calculations, except for IUNI, which is taken from the mathematical subroutine library of the Langley Central Scientific Computer Complex.

```
C
C
C    THIS SUBROUTINE READS DATA GENERATED
C    BY THE PROGRAM SHIELDOSE FOR 1/2 DOSE
C    AT THE CENTER OF A SOLID ALUMINUM SPHERE
C    THE DATA IS FOR SHIELD DEPTHS OF .03 TO 30. (GM/CM2),
C    FOR 200 KM AT 30, 60, AND 90 DEGREES INCLINATION
C    AND FOR 400, 600, 800, AND 1000 KM AT 0, 30, 60,
C    AND 90 DERGREES INCLINATION
C    THE DATA IS CONVERTED BY THE SUBROUTINE TO FULLDOSE
C    FOR RAD/DAY
C
C
      SUBROUTINE READTPE(DOSEF,Z)
      DIMENSION Z(42),DOST(42),DOSEF(42,4,5)
      INC=1
      IORB=1
   10 INC=INC+1
      IF(INC.LT.5) GO TO 20
      INC=1
      IORB=IORB+1
   20 READ(20,100)IMAX
  100 FORMAT(I5)
      IF(EOF(20).NE.0) GO TO 90
      DO 30 I=1,42
      READ(20,200) Z(I),DOST(I)
  200 FORMAT(22X,F11.3,66X,E11.3)
      DOST(I)=2.*DOST(I)
      DOST(I)=DOST(I)/365.
      DOSEF(I,INC,IORB)=DOST(I)
   30 CONTINUE
      GO TO 10
   90 RETURN
      END
C
C
C   THIS SUBROUTINE CALCULATES EXPOSURE RATES TO THE
C   HUMAN BODY FOR VARIOUS BODY ORGANS AND COMPARES
C   THESE RATES TO EXPOSURE LIMITS. THE DOSE RATES ARE
C   CALCULATED AS A FUNCTION OF THE FOLLOWING VARIABLES
C   PASSED TO THE SUBROUTINE:
C
C    1. NDEG
C       DEGREE OF INCLINATION OF ORBIT
C       MUST BE EITHER 0,30,60, OR 90
C
C    2. DIST
C       ALTITUDE OF ORBIT
C       BETWEEN 200. AND 1000. KM
C
```

```
C    3. DAYS
C       NUMBER OF DAYS IN MISSION
C
C    4. ZINPUT
C       THICKNESS OF SPACECRAFT
C       BETWEEN .03 AND 24.5 GM/CM2
C
C
      SUBROUTINE DOSECLC(NDEG,DIST,DAYS,ZINPUT,DRATIO,DTOT)
      DIMENSION DOSEF(42,4,5),Z(42),R(4),C(4),DOST(42)
     1 ,TIME(4),DOSLMT(4,4),DRATIO(4),DTOT(4),ALT(5)
     2 ,DOST1(42),DOST2(42)
      CALL READTPE(DOSEF,Z)
      CALL DOSELMT(DOSLMT,TIME)
      IPT=-1
      INC=NDEG/30+1
      NF=1
      DO 20 I=1,4
      IF(DAYS.GT.TIME(I)) NF=I+1
      IF(NF.EQ.4) NF=3
   20 CONTINUE
      ALT(1)=200.
      DO 30 I=2,5
   30 ALT(I)=ALT(I-1)+200.
      DO 50 I=1,5
      IF(DIST.NE.ALT(I)) GO TO 57
      IORB2=I
      IORB1=IORB2
      GO TO 60
   57 IF(DIST.GT.ALT(I)) IORB1=I
      IORB2=IORB1+1
   50 CONTINUE
   60 IF(INC.NE.1.AND.DIST.GE.400.) GO TO 65
      IORB2=IORB1
      P=(DIST-400.)/100.
      DO 40 I=1,42
      DOST1(I)=(10.**P)*DOSEF(I,1,2)
      DOST1(I)=ALOG(DOST1(I))
      DOST2(I)=DOST1(I)
   40 CONTINUE
      GO TO 75
   65 DO 70 I=1,42
      DOST1(I)=DOSEF(I,INC,IORB1)
      DOST2(I)=DOSEF(I,INC,IORB2)
      DOST1(I)=ALOG(DOST1(I))
      DOST2(I)=ALOG(DOST2(I))
   70 CONTINUE
   75 CALL COEFF(ZINPUT,R,C)
      DO 80 J=1,4
      IF(J.NE.2) GO TO 99
      IF(ZINPUT.LE.8.0) R(J)=ZINPUT/4.
   99 X=ZINPUT+R(J)
      CALL IUNI(42,42,Z,1,DOST1,1,X,DOSTX1,IPT,IRR1)
      CALL IUNI(42,42,Z,1,DOST2,1,X,DOSTX2,IPT,IRR2)
      DOSTX1=EXP(DOSTX1)
      DOSTX2=EXP(DOSTX2)
      DOSTX1=C(J)*DOSTX1
```

```
      DOSTX2=C(J)*DOSTX2
      ANUM=ALOG(DOSTX1/DOSTX2)
      ADEN=ALOG(ALT(IORB1)/ALT(IORB2))
      IF(IORB1.NE.IORB2) A=ANUM/ADEN
      IF(IORB1.EQ.IORB2) A=1.
      ALT1=ALT(IORB1)
      IF(INC.EQ.1.AND.DIST.LT.400.) ALT1=DIST
      B=DOSTX1/(ALT1**A)
      D=B*(DIST**A)
      DTOT(J)=D*DAYS
      IYRS=DAYS/365
      IF(IYRS.EQ.0) IYRS=1
      DRATIO(J)=DTOT(J)/(IYRS*DOSLMT(J,NF))
   80 CONTINUE
      WRITE(5,100) ZINPUT
      WRITE(5,110) DAYS
      WRITE(5,120) DIST,NDEG
      WRITE(5,130)
  100 FORMAT(///,2X,* SHIELD THICKNESS= *,F10.3,* (GM/CM2)*)
  110 FORMAT(/,2X,* DAYS IN THE MISSION *,F9.4)
  120 FORMAT(3X,F6.2,* KM*,I4,* DEGREES*)
  130 FORMAT(/,3X,* DOSE IN RADS *)
      WRITE(5,140)
  140 FORMAT(/,6X,*  B.F.O.        SKIN          LENS           TESTES*)
      WRITE(5,150) (DTOT(I),I=1,4)
  150 FORMAT(/,2X,4E13.5)
      WRITE(5,160)
  160 FORMAT(//,*  FRACTION OF EXPOSURE LIMIT *)
      NNF=TIME(NF)
      WRITE(5,170) NNF
  170 FORMAT(*  FOR*,I4,* DAY MISSION*)
      WRITE(5,140)
      WRITE(5,150) (DRATIO(I),I=1,4)
      RETURN
      END
C
C
C    THIS SUBROUTINE USING THE SPHERICAL
C    SHELL MODEL OF CRITICAL BODY ORGANS
C    OF BILLINGS AND LANGLEY, GENERATES
C    THE COEFFICIENT C(Z) USED IN THE
C    CALCULATION OF THE DOSE TO BODY
C    ORGANS,
C
C       C(Z)=ATAB+BTAB*EXP(-AL*T)
C
C    WHERE T IS THE SHIELD DEPTH
C
C
```

```
      SUBROUTINE COEFF(T,R,C)
      DIMENSION R(4),C(4),RTAB(4),ATAB(4),BTAB(4),AL(4)
      DATA RTAB/5.5,2.,.5,5.5/
      DATA ATAB/.502,.720,.599,.641/
      DATA BTAB/0.,-.356,-.206,.428/
      DATA AL/1.0,.493,.25,.57/
      DO 20 I=1,4
      R(I)=RTAB(I)
   20 C(I)=ATAB(I)+BTAB(I)*EXP(-AL(I)*T)
         RETURN
         END
C
C
C     THIS SUBROUTINE CONTAINS EXPOSURE LIMITS FOR
C     THE B.F.O., SKIN. LENS, AND TESTES FOR 30,90,365,
C     AND 3650 DAY PERIODS
C
C
         SUBROUTINE DOSELMT(DOSE,TIME)
         REAL DOSLMT(4,4),DOSE(4,4),TLMT(4),TIME(4)
         DATA TLMT,DOSLMT/30.,90.,365.,3650.,
        1 25.,75.,37.,13.,30.,80.0,40.,18.,
        2 60.,170.,85.0,38.,200.,600.0,300.,200./
         DO 1 J=1,4
         DO 1 I=1,4
         TIME(J)=TLMT(J)
         DOSE(I,J)=DOSLMT(I,J)
    1    CONTINUE
         RETURN
         END

      SUBROUTINE IUNI(NMAX,N,X,NTAB,Y,IORDER,X0,Y0,IPT,IERR)            FTN41236
C     E1.1                                                              IUNI   3
C***********************************************************************IUNI   4
C*                                                                     *IUNI   5
C*    PURPOSE:                                                         *IUNI   6
C*                SUBROUTINE IUNI USES FIRST OR SECOND ORDER           *IUNI   7
C*                LAGRANGIAN INTERPOLATION TO ESTIMATE THE VALUES      *IUNI   8
C*                OF A SET OF A SET OF FUNCTIONS AT A POINT X0.  IUNI  *IUNI   9
C*                USES ONE INDEPENDENT VARIABLE TABLE AND A DEPENDENT  *IUNI  10
C*                VARIABLE TABLE FOR EACH FUNCTION TO BE EVALUATED.    *IUNI  11
C*                THE ROUTINE ACCEPTS THE INDEPENDENT VARIABLES SPACED *IUNI  12
C*                AT EQUAL OR UNEQUAL INTERVALS.  EACH DEPENDENT       *IUNI  13
```

```
C*                    VARIABLE TABLE MUST CONTAIN FUNCTION VALUES CORRES-  *IUNI
C*                    PONDING TO EACH X(I) IN THE INDEPENDENT VARIABLE     *IUNI
C*                    TABLE.  THE ESTIMATED VALUES ARE RETURNED IN THE YO  *IUNI
C*                    ARRAY WITH THE N-TH VALUE OF THE ARRAY HOLDING THE   *IUNI
C*                    VALUE OF THE N-TH FUNCTION VALUE EVALUATED AT X0.    *IUNI
C*                                                                         *IUNI
C*       USE:                                                              *IUNI
C*                    CALL IUNI(NMAX,N,X,NTAB,Y,IORDER,X0,Y0,IPT,IERR)     *IUNI
C*                                                                         *IUNI
C* PARAMETERS:                                                             *IUNI
C*                                                                         *IUNI
C*       NMAX         THE MAXIMUM NUMBER OF POINTS IN THE INDEPENDENT      *IUNI
C*                    VARIABLE ARRAY.                                      *IUNI
C*                                                                         *IUNI
C*          N         THE ACTUAL NUMBER OF POINTS IN THE INDEPENDENT       *IUNI
C*                    ARRAY,WHERE N .LE. NMAX.                             *IUNI
C*                                                                         *IUNI
C*          X         A ONE-DIMENSIONAL ARRAY, DIMENSIONED (NMAX) IN THE   *IUNI
C*                    CALLING PROGRAM, WHICH CONTAINS THE INDEPENDENT      *IUNI
C*                    VARIABLES.  THESE VALUES MUST BE STRICTLY MONOTONIC. *IUNI
C*                                                                         *IUNI
C*       NTAB         THE NUMBER OF DEPENDENT VARIABLE TABLES              *IUNI
C*                                                                         *IUNI
C*          Y         A TWO-DIMENSIONAL ARRAY DIMENSIONED (NMAX,NTAB) IN   *IUNI
C*                    THE CALLING PROGRAM.  EACH COLUMN OF THE ARRAY       *IUNI
C*                    CONTAINS A DEPENDENT VARIABLE TABLE                  *IUNI
C*                                                                         *IUNI
C*     IORDER         INTERPOLATION PARAMETER SUPPLIED BY THE USER.        *IUNI
C*                                                                         *IUNI
```

```
C*            =0  ZERO ORDER INTERPOLATION: THE FIRST FUNCTION      *IUNI  43
C*                VALUE IN EACH DEPENDENT VARIABLE TABLE IS         *IUNI  44
C*                ASSIGNED TO THE CORRESPONDING MEMBER OF THE Y0    *IUNI  45
C*                ARRAY.  THE FUNCTIONAL VALUE IS ESTIMATED TO      *IUNI  46
C*                REMAIN CONSTANT AND EQUAL TO THE NEAREST KNOWN    *IUNI  47
C*                FUNCTION VALUE.                                   *IUNI  48
C*                                                                  *IUNI  49
C*       X0   THE INPUT POINT AT WHICH INTERPOLATION WILL BE        *IUNI  50
C*            PERFORMED.                                            *IUNI  51
C*                                                                  *IUNI  52
C*       Y0   A ONE-DIMENSIONAL ARRAY DIMENSIONED (NTAB) IN THE     *IUNI  53
C*            CALLING PEOGRAM.  UPON RETURN THE ARRAY CONTAINS THE  *IUNI  54
C*            ESTIMATED VALUE OF EACH FUNCTION AT X0.               *IUNI  55
C*                                                                  *IUNI  56
C*      IPT   ON THE FIRST CALL IPT MUST BE INITIALIZED TO -1 SO    *IUNI  57
C*            THAT MONOTONICITY WILL BE CHECKED. UPON LEAVING THE   *IUNI  58
C*            ROUTINE IPT EQUALS THE VALUE OF THE INDEX OF THE X    *IUNI  59
C*            VALUE PRECEDING X0 UNLESS EXTRAPOLATION WAS           *IUNI  60
C*            PERFORMED.  IN THAT CASE THE VALUE OF IPT IS          *IUNI  61
C*            RETURNED AS:                                          *IUNI  62
C*            =0  DENOTES X0 .LT. X(1) IF THE X ARRAY IS IN         *IUNI  63
C*                INCREASING ORDER AND X(1) .GT. X0 IF THE X ARRAY  *IUNI  64
C*                IS IN DECREASING ORDER.                           *IUNI  65
C*            =N  DENOTES X0 .GT. X(N) IF THE X ARRAY IS IN         *IUNI  66
C*                INCREASING ORDER AND X0 .LT. X(N) IF THE X ARRAY  *IUNI  67
C*                IS IN DECREASING ORDER.                           *IUNI  68
C*                                                                  *IUNI  69
C*            ON SUBSEQUENT CALLS, IPT IS USED AS A POINTER TO      *IUNI  70
C*            BEGIN THE SEARCH FOR X0.                              *IUNI  71
```

```
C*                                                                      *IUNI  72
C*       IERR    ERROR PARAMETER GENERATED BY THE ROUTINE               *IUNI  73
C*               =0  NORMAL RETURN                                      *IUNI  74
C*               =J  THE J-TH ELEMENT OF THE X ARRAY IS OUT OF ORDER    *IUNI  75
C*               =-1 ZERO ORDER INTERPOLATION PERFORMED BECAUSE         *IUNI  76
C*                   IORDER =0.                                         *IUNI  77
C*               =-2 ZERO ORDER INTERPOLATION PERFORMED BECAUSE ONLY    *IUNI  78
C*                   ONE POINT WAS IN X ARRAY.                          *IUNI  79
C*               =-3 NO INTERPOLATION WAS PERFORMED BECAUSE             *IUNI  80
C*                   INSUFFICIENT POINTS WERE SUPPLIED FOR SECOND       *IUNI  81
C*                   ORDER INTERPOLATION.                               *IUNI  82
C*               =-4 EXTRAPOLATION WAS PERFORMED                        *IUNI  83
C*                                                                      *IUNI  84
C*               UPON RETURN THE PARAMETER IERR SHOULD BE TESTED IN     *IUNI  85
C*               THE CALLING PROGRAM.                                   *IUNI  86
C*                                                                      *IUNI  87
C*   REQUIRED ROUTINES                NONE                              *IUNI  88
C*                                                                      *IUNI  89
C*   SOURCE                           CMPB ROUTINE MTLUP MODIFIED       *IUNI  90
C*                                    BY COMPUTER SCIENCES CORPORATION*IUNI  91
C*                                                                      *IUNI  92
C*   LANGUAGE                         FORTRAN                           *IUNI  93
C*                                                                      *IUNI  94
C*                                                                      *IUNI  95
C*   DATE RELEASED                    AUGUST 1,1973                     *IUNI  96
C*                                                                      *IUNI  97
C*   LATEST REVISION                  AUGUST 1,1973                     *IUNI  98
C*                                                                      *IUNI  99
C***********************************************************************IUNI 100
```

```
      DIMENSION X(*),Y(NMAX,*),Y0(*)                                    FTN41237
      NM1=N-1                                                           IUNI 105
      IERR=0                                                            IUNI 106
      J=1                                                               IUNI 107
      DELX=X(2)-X(1)                                                    FTN41239
C                                                                       IUNI 109
C               TEST FOR ZERO ORDER INTERPOLATION                       IUNI 110
C                                                                       IUNI 111
      IF (IORDER .EQ. 0) GO TO 10                                       IUNI 112
      IF (N.LT. 2) GO TO 20                                             IUNI 113
      GO TO 50                                                          IUNI 114
   10 IERR=-1                                                           IUNI 115
      GO TO 30                                                          IUNI 116
   20 IERR=-2                                                           IUNI 117
   30 DO 40 NT=1,NTAB                                                   IUNI 118
         Y0(NT)=Y(1,NT)                                                 IUNI 119
   40    CONTINUE                                                       IUNI 120
      RETURN                                                            IUNI 121
   50 IF (IPT .GT. -1) GO TO 65                                         IUNI 122
C                                                                       IUNI 123
C             CHECK FOR TABLE OF NODE POINTS BEING STRICTLY MONOTONIC   IUNI 124
C             THE SIGN OF DELX SIGNIFIES WHETHER TABLE IS IN            IUNI 125
C             INCREASING OR DECREASING ORDER.                           IUNI 126
C                                                                       IUNI 127
      IF (DELX .EQ. 0) GO TO 190                                        IUNI 128
      IF (N .EQ. 2) GO TO 65                                            IUNI 129
C                                                                       IUNI 130
C             CHECK FOR SIGN CONSISTENCY IN THE DIFFERENCES OF          IUNI 131
C             SUBSEQUENT PAIRS                                          IUNI 132
```

```
C
      DO 60 J=2,NM1
         IF (DELX * (X(J+1)-X(J))) 190,190,60
   60    CONTINUE
C
C                IPT IS INITTALIZED TO BE WITHIN THE INTERVAL
C
   65 IF (IPT .LT. 1) IPT=1
      IF (IPT .GT. NM1) IPT=NM1
      IN= SIGN (1.0,DELX *( X0-X(IPT)))
   70 P= X(IPT) - X0
      IF (P* (X(IPT +1)- X0)) 90,180,80
   80 IPT =IPT +IN
C
C                TEST TO SEE IF IT IS NECCESARY TO EXTRAPOLATE
C
      IF (IPT.GT.0 .AND. IPT .LT. N) GO TO 70
      IERR=-4
      IPT=IPT- IN
C
C                TEST FOR ORDER OF INTERPOLATION
C
C
   90 IF (IORDER .GT. 1) GO TO 120
C
C                FIRST ORDER INTERPOLATION
C
      DO 100 NT=1,NTAB
        YO(NT)=Y(IPT,NT)+((Y(IPT+1,NT)- Y(IPT,NT))*(X0-X(IPT)))/
```

```
     1            (X(IPT+1)-X(IPT))
 100     CONTINUE
      IF (IERR .EQ. -4) IPT=IPT+IN
      RETURN
C
C           SECOND ORDER INTERPOLATION
C
 120  IF (N .EQ. 2) GO TO 200
C
C           CHOOSING A THIRD POINT SO AS TO MINIMIZE THE DISTANCE
C           BETWEEN THE THREE POINTS USED TO INTERPOLATE
C
      IF (IPT .EQ. NM1) GO TO  140
      IF (IPT .EQ. 1) GO TO 130
      IF (DELX *(X0-X(IPT-1)).LT.DELX* (X(IPT+2)-X0)) GO TO 140
 130  L=IPT
      GO TO 150
 140  L=IPT -1
 150  V1=X(L)-X0
      V2=X(L+1)-X0
      V3=X(L+2)-X0
      DO 160 NT=1,NTAB
      YY1=(Y(L,NT) * V2 - Y(L+1,NT) * V1)/(X(L+1) - X(L))
      YY2=(Y(L+1,NT)*V3-Y(L+2,NT) *V2)/(X(L+2)-X(L+1))
 160  Y0(NT)=(YY1*V3 -YY2*V1)/(X(L+2)-X(L))
      IF (IERR .EQ. -4) IPT=IPT + IN
      RETURN
 180  IF(P .NE. 0) IPT=IPT +1
      DO 185 NT=1,NTAB
```

```
      YO(NT)=Y(IPT,NT)
 185  CONTINUE
      RETURN
C
C            IERR IS SET TO THE SUBSCRIPT OF THE MEMBER OF THE TABLE
C            WHICH IS OUT OF ORDER
C
 190  IERR=J +1
      RETURN
 200  IERR=-3
      RETURN
      END
```

Appendix B

Example of Main Program for a Mission of 90 Days With EVA Occurring 8 Hours Every 10 Days

```
      DIMENSION Z(42),DOST(42),DOSE(4),R(4),C(4),RTAB(4),AL(4)
     1 ,ATAB(4),BTAB(4),TLMT(4),TIME(4),DOSLMT(4,4),DOSET(4,4)
     2 ,DOSEF(42,4,5),ALT(5),DOST1(42),DOST2(42),DRATIO(4),DTOT(4)
     3 ,DTAB(4,6,4),DT(6,4),D1(4),D2(4),D3(4),DR1(4)
     4 ,DR2(4),DR3(4),DRTOT(4)
      NDEG=30
      DIST=500.
      DAYS=90.
      DEVA=8./24.
      FEVA=10.
      TEVA=DEVA*DAYS/FEVA
      DMT=DAYS-TEVA
      Z1=1.0
      Z2=0.2
      Z3=1.0
      CALL DOSECLC(NDEG,DIST,DMT,Z1,D1,DR1)
      CALL DOSECLC(NDEG,DIST,TEVA,Z2,D2,DR2)
      CALL DOSECLC(NDEG,DIST,TEVA,Z3,D3,DR3)
      DO 10 I=1,4
      DTOT(I)=D1(I)+D2(I)
      IF(I.EQ.3) DTOT(I)=D1(I)+D3(I)
      DRTOT(I)=DR1(I)+DR2(I)
      IF(I.EQ.3) DRTOT(I)=DR1(I)+DR3(I)
 10   CONTINUE
      WRITE(5,100)
      WRITE(5,110)
      WRITE(5,115) Z1
      WRITE(5,120) NDEG,DIST
100   FORMAT(//,2X,* RADIATION DOSES AND FRACTION OF EXPOSURE*)
110   FORMAT(2X,* LIMIT FOR 90 DAY MISSION WITH EVA*)
115   FORMAT(/,3X,* SHIELD THICKNESS = *,F5.2,*,GM/CM2*)
120   FORMAT(3X,I3,* DEG INC*,2X,F6.2,*,KM*)
      WRITE(5,130)
130   FORMAT(//,2X,* DOSE, RADS*)
      WRITE(5,140)
140   FORMAT(/,5X,* B.F.O.    SKIN     LENS     TESTES*)
      WRITE(5,150) (DTOT(I),I=1,4)
150   FORMAT(/,2X,4F9.3)
      WRITE(5,160)
160   FORMAT(//,2X,* FRACTION OF EXPOSURE LIMIT*)
      WRITE(5,140)
      WRITE(5,150) (DRTOT(I),I=1,4)
      STOP
      END
```

References

1. Alsmiller, R. G., Jr.; Irving, D. C.; Kinney, W. E.; and Moran, H. S.: The Validity of the Straightahead Approximation in Space Vehicle Shielding Studies. *Second Symposium on Protection Against Radiations in Space*, Arthur Reetz, Jr., ed., NASA SP-71, 1965, pp. 177–181.
2. Jordan, T. M.; Koprowski, E. F.; and Langley, R. W.: Shielding Requirements for Manned Orbiting Space Stations. *Second Symposium on Protection Against Radiations in Space*, Arthur Reetz, Jr., ed., NASA SP-71, 1965, pp. 415–427.
3. Jordan, T. M.: *Electron Dose Attenuation Kernels for Slab and Spherical Geometries.* AFWL-TR-81-43, U.S. Air Force, Nov. 1981. (Available from DTIC as AD A115 232.)
4. Kase, Paul G.: Influence of a Detailed Model of Man on Proton Depth/Dose Calculations. *Proceedings of National Symposium on Natural and Manmade Radiation in Space*, E. A. Warman, ed., NASA TM X-2440, 1972, pp. 773–780.
5. Wilson, John W.; and Cucinotta, Frank: *Human Exposure in Low Earth Orbit.* NASA TP-2344, 1984.
6. Stassinopoulos, E. G.: *Space Radiation Incident on SATS Missions.* NASA TM X-70544, 1973.
7. Seltzer, Stephen: *SHIELDOSE: A Computer Code for Space-Shielding Radiation Dose Calculations.* NBS Tech. Note 1116, U.S. Dep. Commer., May 1980.
8. Billings, M. P.; and Yucker, W. R.: *The Computerized Anatomical Man (CAM) Model.* NASA CR-134043, 1973.
9. Billings, M. P.; and Langley, R. W.: *Monitoring of Space Proton Dose to Body Organs.* MDAC Paper WD 2355, McDonnell Douglas Astronautics Co., July 1974.
10. Space Science Board: *Radiation Protection Guides and Constraints for Space-Mission and Vehicle-Design Studies Involving Nuclear Systems.* Nat. Acad. Sci.—Nat. Res. Counc., 1970.

TABLE 1. HUMAN BODY GEOMETRY PARAMETERS USED IN PRESENT CALCULATIONS

Organ	r, g/cm^2	a	b	α
B.F.O.	5.5	0.502	0.000	1.0
Testes	5.5	.641	.428	.57
Lens	.5	.599	−.206	.25
Skin*	$z/4$	.720	−.356	.493

*$r \leq 2$ g/cm^2.

TABLE 2. RELEVANT VALUES OF SHIELD THICKNESS

z, g/cm^2 of Al	Place of occurrence
0.2	Space suit
1.0	Space helmet, Skylab wall
2.0	Heavily shielded habitat
5.0	Heavily shielded vehicle, solar cosmic ray shelter

TABLE 3. SUGGESTED EXPOSURE LIMITS AND EXPOSURE ACCUMULATION RATE CONSTRAINTS FOR UNIT REFERENCE RISK CONDITIONS

Constraint	Primary reference risk, rems[1] at 5 cm	Ancillary reference risks			
		Bone marrow, rems at 5 cm	Skin, rems at 0.1 mm	Ocular lens, rems at 3 mm	Testes,[2] rems at 3 cm
1-year average daily rate		0.2	0.6	0.3	0.1
30-day minimum		25.0	75.0	37.0	13.0
Quarterly maximum		30.0	80.0	40.0	18.0
Yearly maximum		60.0	170.0	85.0	38.0
Career limit	400	200.0	600.0	300.0	200.0

[1]Rem = Radiation absorbed dose in rads times a quality factor q to account for the different relative biological effectiveness (RBE) of different radiations ($q = 1.2$).

[2]Values taken from reference 10.

TABLE 4. SAMPLE CALCULATIONS FOR FRACTION OF EXPOSURE LIMIT FOR 90-DAY MISSION

Shield thickness, g/cm^2 of Al	Altitude, km	Fraction of exposure limit for 90-day mission for—			
		B.F.O.	Skin	Lens	Testes
0° inclined orbits					
1.00	425	0.015	0.005	0.009	0.043
	450	.028	.010	.017	.081
	475	.052	.018	.032	.152
	500	.097	.035	.061	.285
	525	.182	.065	.113	.533
	550	.338	.120	.211	.992
1.50	425	.015	.006	.009	.041
	450	.028	.011	.018	.076
	475	.053	.020	.033	.144
	500	.098	.037	.062	.269
	525	.184	.070	.116	.502
	550	.342	.130	.216	.935
2.00	425	.015	.006	.010	.039
	450	.028	.011	.018	.073
	475	.053	.021	.035	.137
	500	.099	.040	.065	.256
	525	.185	.075	.121	.479
	550	.345	.139	.225	.892
30° inclined orbits					
1.00	425	0.232	0.312	0.476	0.681
	450	.285	.384	.586	.836
	475	.346	.468	.713	1.015
	500	.416	.564	.858	1.221
	525	.496	.675	1.024	1.454
	550	.586	.800	1.211	1.718
1.50	425	.220	.224	.356	.601
	450	.271	.272	.431	.739
	475	.329	.327	.517	.899
	500	.396	.388	.614	1.082
	525	.473	.457	.722	1.291
	550	.559	.535	.844	1.528
2.00	425	.203	.200	.323	.525
	450	.250	.242	.391	.647
	475	.305	.290	.468	.788
	500	.368	.344	.555	.950
	525	.440	.404	.653	1.136
	550	.521	.472	.762	1.346

TABLE 4. Concluded

Shield thickness, g/cm^2 of Al	Altitude, km	Fraction of exposure limit for 90-day mission for—			
		B.F.O.	Skin	Lens	Testes
60° inclined orbits					
1.00	425	0.156	0.884	0.715	0.458
	450	.185	.968	.796	.544
	475	.218	1.054	.881	.640
	500	.255	1.143	.970	.747
	525	.295	1.235	1.064	.865
	550	.339	1.329	1.161	.995
1.50	425	.147	.233	.283	.402
	450	.175	.266	.328	.479
	475	.206	.301	.378	.564
	500	.241	.339	.433	.659
	525	.280	.380	.491	.764
	550	.322	.423	.555	.880
2.00	425	.135	.146	.236	.349
	450	.161	.170	.275	.416
	475	.190	.197	.318	.491
	500	.223	.226	.365	.575
	525	.259	.258	.417	.668
	550	.299	.293	.473	.771
90° inclined orbits					
1.00	425	0.129	0.771	0.609	0.378
	450	.153	.840	.677	.450
	475	.181	.912	.747	.530
	500	.211	.985	.821	.619
	525	.245	1.060	.898	.717
	550	.282	1.137	.979	.826
1.50	425	.122	.194	.231	.332
	450	.145	.221	.269	.396
	475	.171	.251	.310	.467
	500	.200	.282	.355	.546
	525	.232	.316	.405	.634
	550	.268	.352	.458	.731
2.00	425	.112	.120	.193	.288
	450	.133	.140	.226	.344
	475	.158	.162	.262	.407
	500	.185	.187	.301	.477
	525	.215	.213	.345	.555
	550	.248	.242	.391	.641

TABLE 5. DOSE TO CRITICAL BODY ORGANS

Shield thickness, g/cm^2 of Al	Altitude, km	Dose, rad B.F.O.	Dose, rad Skin	Dose, rad Lens	Dose, rad Testes
0° inclined orbits					
1.00	425	0.441	0.419	0.366	0.776
	450	.831	.788	.690	1.461
	475	1.559	1.479	1.295	2.743
	500	2.919	2.769	2.424	5.135
	525	5.450	5.170	4.525	9.588
	550	10.154	9.632	8.431	17.861
1.50	425	.446	.452	.375	.731
	450	.840	.851	.707	1.377
	475	1.576	1.597	1.327	2.584
	500	2.950	2.989	2.484	4.837
	525	5.509	5.581	4.637	9.032
	550	10.263	10.398	8.639	16.826
2.00	425	.450	.484	.391	.697
	450	.848	.912	.736	1.313
	475	1.591	1.711	1.381	2.465
	500	2.978	3.203	2.586	4.614
	525	5.560	5.981	4.828	8.616
	550	10.358	11.142	8.995	16.051
30° inclined orbits					
1.00	425	6.970	24.923	19.057	12.261
	450	8.557	30.716	23.437	15.053
	475	10.390	37.429	28.504	18.276
	500	12.490	45.150	34.319	21.970
	525	14.880	53.967	40.948	26.174
	550	17.583	63.972	48.458	30.930
1.50	425	6.603	17.959	14.251	10.825
	450	8.119	21.773	17.253	13.311
	475	9.872	26.124	20.673	16.185
	500	11.884	31.052	24.542	19.484
	525	14.178	36.600	28.893	23.244
	550	16.775	42.810	33.757	27.503
2.00	425	6.098	16.023	12.936	9.449
	450	7.513	19.378	15.644	11.642
	475	9.153	23.195	18.726	14.183
	500	11.039	27.509	22.209	17.105
	525	13.191	32.355	26.121	20.441
	550	15.634	37.769	30.492	24.226

TABLE 5. Concluded

Shield thickness, g/cm^2 of Al	Altitude, km	Dose, rad B.F.O.	Skin	Lens	Testes
60° inclined orbits					
1.00	425	4.685	70.750	28.583	8.242
	450	5.565	77.436	31.832	9.788
	475	6.548	84.340	35.243	11.518
	500	7.641	91.459	38.817	13.440
	525	8.849	98.787	42.553	15.566
	550	10.178	106.320	46.449	17.904
1.50	425	4.417	18.614	11.317	7.241
	450	5.254	21.248	13.139	8.614
	475	6.192	24.082	15.133	10.151
	500	7.236	27.120	17.302	11.863
	525	8.391	30.364	19.655	13.757
	550	9.665	33.817	22.194	15.845
2.00	425	4.056	11.672	9.423	6.286
	450	4.835	13.620	10.996	7.493
	475	5.709	15.762	12.725	8.847
	500	6.684	18.104	14.616	10.357
	525	7.765	20.654	16.675	12.033
	550	8.958	23.419	18.907	13.882
90° inclined orbits					
1.00	425	3.868	61.673	24.371	6.804
	450	4.600	67.225	27.071	8.092
	475	5.420	72.937	29.900	9.533
	500	6.332	78.804	32.856	11.138
	525	7.341	84.821	35.939	12.914
	550	8.454	90.986	39.148	14.871
1.50	425	3.645	15.507	9.232	5.976
	450	4.342	17.702	10.746	7.119
	475	5.124	20.063	12.406	8.401
	500	5.995	22.594	14.217	9.829
	525	6.961	25.297	16.185	11.413
	550	8.027	28.175	18.314	13.160
2.00	425	3.348	9.567	7.724	5.188
	450	3.997	11.190	9.034	6.193
	475	4.725	12.978	10.477	7.322
	500	5.539	14.937	12.059	8.583
	525	6.442	17.075	13.785	9.983
	550	7.440	19.397	15.660	11.530

TABLE 6. RADIATION DOSES AND FRACTION OF EXPOSURE LIMIT FOR 90-DAY MISSION WITH EVA

[Shield thickness = 1.00 g/cm^2; 30° inclined orbits; h = 500 km]

	B.F.O.	Skin	Lens	Testes
Dose, rad	12.545	54.640	34.319	22.198
Fraction of exposure limit	0.421	0.692	0.860	1.254

1. Report No. NASA TM-86324	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle COMPUTER SUBROUTINES FOR ESTIMATION OF HUMAN EXPOSURE TO RADIATION IN LOW EARTH ORBIT		5. Report Date February 1985
		6. Performing Organization Code 506-53-23-03
7. Author(s) Francis A. Cucinotta and John W. Wilson		8. Performing Organization Report No. L-15865
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665		10. Work Unit No.
		11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		13. Type of Report and Period Covered Technical Memorandum
		14. Sponsoring Agency Code

15. Supplementary Notes

Francis A. Cucinotta: Old Dominion University, Norfolk, Virginia.
John W. Wilson: Langley Research Center, Hampton, Virginia.

16. Abstract

Computer subroutines to calculate human exposure to trapped radiations in low Earth orbit (LEO) on the basis of a simple approximation of the human geometry by spherical shell shields of varying thickness are presented and detailed. The subroutines calculate the dose to critical body organs and the fraction of exposure limit reached as a function of altitude of orbit, degree of inclination, shield thickness, and days in mission. Exposure rates are compared with current exposure limits.

17. Key Words (Suggested by Authors(s)) Space operations Radiation protection Trapped radiation	18. Distribution Statement Unclassified—Unlimited Subject Category 54

19. Security Classif.(of this report) Unclassified	20. Security Classif.(of this page) Unclassified	21. No. of Pages 24	22. Price A02

For sale by the National Technical Information Service, Springfield, Virginia 22161

NASA-Langley, 1985

National Aeronautics and
Space Administration

Washington, D.C.
20546

Official Business
Penalty for Private Use, $300

THIRD-CLASS BULK RATE

Postage and Fees Paid
National Aeronautics and
Space Administration
NASA-451

U.S.MAIL

NASA

POSTMASTER: If Undeliverable (Section 158 Postal Manual) Do Not Return